

Dokumentáció az 1. feladatsorhoz

(egyszerű, rövidített kivitelben)

Felhasználói dokumentáció

Feladat:

Adjuk meg két N elemű vektor skalárszorzatát!

$$\text{Skalárszorzat} : \langle X, Y \rangle := \sum_{i=1}^N x_i * y_i$$

Környezet:

IBM PC, legalább MS DOS 5.0 operációs rendszer.

Használat:

A program indítása:

A program SKALAR.EXE néven található az A:\SKALAR\ könyvtárban, a neve leírásával indítható.

Ha az input a be.txt fájlban, van akkor indítása:

SKALAR.EXE be.txt

A program bemenete:

Ha az input *fájlban* van, akkor az adatokat *külön-külön sorban* keresi a program. Mivel a fájl és a klaviatúra-input könnyen megfeleltethető egymásnak, ezért ez esetben nem részletezzük a fájl esetet.

A program első paramétere az vektorok elemszáma, majd egyesével az első vektor elemei következnek, utána a második vektoré.

Egy lehetséges párbeszéd (*vastagon és dőlten* szedtük a felhasználó válaszait):

```
Skalárszorzás

Hány eleműek a vektorok (maximum 100)?3

Kérem egyesével az 1. vektor elemeit!

1. elem: 5
2. elem: 2.5
3. elem: 3

Kérem egyesével a 2. vektor elemeit!

1. elem: 1
2. elem: 4
3. elem: 3
```

A program eredménye:

A program visszaírja az általunk beírt két vektor elemeit, majd megadja a skalárszorzatukat.

Egy lehetséges kimenet:

```
Skalárszorzás

1. vektor: 5.0, 2.5, 3.0
2. vektor: 1.0, 4.0, 3.0

Skalárszorzat: 24.0

Nyomj meg egy billentyűt!
```

Hibalehetőségek:

A vektorok elemszáma nem lehet több 100-nál. Ha a felhasználó ennél többet, vagy 0-nál kevesebbet ad meg, akkor a program az elemszámot újra kérdezi.

Egy lehetséges párbeszéd:

```
Skalárszorzás

Hány eleműek a vektorok (maximum 100)?123
Hány eleműek a vektorok (maximum 100)?-5
Hány eleműek a vektorok (maximum 100)?4
Kérem egyesével az 1. vektor elemeit!
1. elem: |
```

Mivel a fájl input esetén nincs lehetőség újrakérdezésre, a program feltételezi, hogy ekkor az adatok helyesek formailag és tartalmilag is.

Fejlesztői dokumentáció

Feladat:

Adjuk meg két N elemű vektor skalárszorzatát!

Specifikáció:

Bemenet: $N \in \mathbb{N}$ [a két vektor elemszáma, maximális értéke 100]
 $X, Y \in \mathbb{R}^*$ [tetszőleges valós számokat tartalmazó vektorok]

Kimenet: $XY \in \mathbb{R}$ [a két vektor skalárszorzata]

Előfeltétel: $N = \text{Hossz}(X) = \text{Hossz}(Y)$
 [fölvethető kérdés: értelmes-e $N=0$ esetén]

Utófeltétel: $XY = \langle X, Y \rangle$

(Fogalom)definíció:

$$\langle X, Y \rangle := \sum_{i=1}^N x_i * y_i$$

Környezet:

IBM PC, legalább MS DOS 5.0 operációs rendszer,
 Free Pascal (Geany) fejlesztői környezet, Crt unit.¹

Forráskód:

A program SKALAR.PAS néven található a A: \FORRAS nevű könyvtárban.
 (A biztonsági másolat helye: A: \MASOLAT)

Megoldás:

Fontos típusok, változók:

Konstans

MaxN: Egész (100) – a megoldásban használt vektorok
 – maximális mérete

Típus

Vektor = Tömb (1..MaxN: Valós) – a megoldásban használt vektorok
 – típusa

Változó

N: Egész
 X, Y: Vektor
 XY: Valós

(A bemeneten és a kimeneten kívül nincs fontos változó a megoldásban.)

Algoritmus:

```

Program SkalárisSzorzat:
  Adatbeolvasás(N, X, Y)
  SkalárSzorzat(N, X, Y, XY)
  EredményKiírás(N, X, Y, XY)
Program vége.
  
```

¹ esetleg Turbo Pascal 5.5. vagy későbbi verzió

```

Eljárás AdaBeolvasás(Változó N:Egész, X,Y:Vektor):
  Be: N [N>0 és N≤MaxN]
  Be: X
  Be: Y
Eljárás vége.

Eljárás SkalarSzorzat(Konstans N:Egész, X,Y:Vektor,
  Változó XY:Valós):
  Változó i: Egész
  XY:=0
  Ciklus i=1-től N-ig
    XY:=XY+X(i)*Y(i)
  Ciklus vége
Eljárás vége.

Eljárás EredményKiírás(Konstans N:Egész, X,Y:Vektor,
  XY:Valós):
  Ki: X,Y
  Ki: XY
Eljárás vége.

```

Pascal kód:

```

Program SkalarSzorzas;
{A program készítője: X Y
 csoportja: Y}

Uses Crt;
Const MaxN=100; {A tömb maximális mérete}
Type Vektor=Array [1..MaxN] of Real;
Var N: Integer; {a vektorok tényleges mérete}
    X,Y: Vektor; {a vektorok}
    XY: Real; {a skáláris szorzat}

Procedure AdatBeolvasas(Var N: Integer; Var X,Y: Vektor);
  Var i: Integer;
Begin
  ClrScr;
  Writeln('Skalárszorzás':46 {középreigazítás}); Writeln;
  Repeat
    {$i-}
    Write('Hány eleműek a vektorok (maximum ',MaxN, ')?');
    Readln(N);
    {$i+}
  Until (IOResult=0) and (N in [1..MaxN]);
  Writeln;
  Writeln('Kérem egyesével az 1. vektor elemeit!'); Writeln;
  For i:=1 to N do
    Begin
      Write(i:5,'. elem:'); Readln(X[i]);
    End; Writeln;
  Writeln('Kérem egyesével a 2. vektor elemeit!'); Writeln;
  For i:=1 to N do
    Begin
      Write(i:5,'. elem:'); Readln(Y[i]);
    End;
End; {AdatBeolvasás}

Procedure SkalarSzorzat(N: Integer; X,Y: Vektor; Var XY: Real);
  Var i: Integer;
Begin
  XY:=0;
  For i:=1 to N do XY:=XY+X[i]*Y[i]
End; {SkalarSzorzat}

```

```

Procedure EredmenyKiiras(N: Integer; X,Y: Vektor; XY: Real);
  Var i: Integer;
Begin
  ClrScr;
  Writeln('Skalárszorzás':46 {középreigazítás});
  Write('1. vektor:');
  For i:=1 to N do Write(X[i]:4:1); Writeln; writeln;
  Write('2. vektor:');
  For i:=1 to N do Write(Y[i]:4:1); Writeln; writeln; Writeln;
  Writeln('Skalárszorzat:', XY:4:1);
  GotoXY(24,20); Write('Nyomj meg egy billentyűt!');
End; {EredményMegjelenítés}

Begin
  AdatBeolvasas(N,X,Y);
  SkalarSzorzat(N,X,Y,XY);
  EredmenyKiiras(N,X,Y,XY);
End.

```

Tesztelés:*Érvényes tesztesetek:*

N=0	XY=0
N=1, X=(1), Y=(1)	XY=1
N=1, X=(4), Y=(2.5)	XY=10
N=4, X=(1,2,3,4), Y=(1,2,3,4)	XY=30
N=100, X=(1,2,...,100), Y=(1,1,...,1)	XY=1275

Érvénytelen tesztesetek:

N=-1	nincs
N=101	nincs

Hatékonyági tesztesetek:

{Egyelőre nincsenek.}

Fejlesztési lehetőségek:

A vektorok elemei nagyobb pontosságú valós számokra kicserélhetők.

Készítő(k):

A programozás módszertan tárgy oktató(i).

EHA-kód (ok)